# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Alexander Baldeck <alexander@archlinux.org>

pkgname=libice
pkgver=1.0.9
pkgrel=1
pkgdesc="X11 Inter-Client Exchange library"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
depends=('glibc' 'xproto>=7.0.18')
makedepends=('xtrans>=1.2.5' 'xorg-util-macros')
source=(${url}/releases/individual/lib/libICE-${pkgver}.tar.bz2{,.sig})
license=('custom')
sha256sums=('8f7032f2c1c64352b5423f6b48a8ebdc339cc63064af34d66a6c9aa79759e202'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd libICE-${pkgver}
  ./configure --prefix=/usr --sysconfdir=/etc
  make
}

package() {
  cd libICE-${pkgver}
  make DESTDIR="${pkgdir}" install

  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
